Comput. Biol. Med. Vol. 17, No. 2, pp. 85–99, 1987.
Printed in Great Britain.

0010–4825/87 $3.00+ .00

FRIEDMAN AND QUADE TESTS: BASIC COMPUTER PROGRAM TO PERFORM NONPARAMETRIC TWO-WAY ANALYSIS OF VARIANCE AND MULTIPLE COMPARISONS ON RANKS OF SEVERAL RELATED SAMPLES

ELVAR THEODORSSON-NORHEIM
Department of Clinical Chemistry, Karolinska Hospital, S-104 01 Stockholm, Sweden

(*Received* 13 *September* 1985; *in revised form* 29 *September* 1986)

Abstract—Friedman and Quade tests are nonparametric two-way analyses of variance which may be used to compare several related samples. The Friedman test is a multisample extension of the sign test while the Quade test is an extension of the Wilcoxon signed-rank for paired samples to the case of several related samples. The Quade test may be more powerful for a small number of treatments while the Friedman test may be more powerful when the number of treatments is five or more. The present program is written in an elementary subset of BASIC and will perform the Friedman and Quade tests quickly on practically every computer programmable in BASIC. It may i.e. be used to analyze biomedical data where the response of experimental subjects to a stimulus is monitored at time intervals.

BASIC Microcomputer Friedman test Quade test Statistics
Nonparametric statistics Multiple comparisons

INTRODUCTION

In many biomedical experiments, a certain property of the same experimental subject is observed on one or more occasions before, during and after a stimulus. Frequently the stimulus does not leave residual effects which means that the observations on a single subject are related. Thus each subject serves as its own control (repeated measurements design). The sensitivity of the experiment is increased in this way because the error caused by the difference between subjects is eliminated, and we are left with the study of the stimulus effect and the experimental error. If the response is short lasting and appears quickly or at a predictable time after the stimulus, it may be sufficient to compare the prestimulatory observations with the stimulated observations by means of the paired t-test or by its nonparametric analogue, the Wilcoxon test. However, if several observations are made during the course of the experiment and the time course of the stimulatory effect is unknown, the experimenter may be interested in estimating when the response observations differ significantly from the basal observations, and when they are likely to have returned to the basal levels. More than four decades ago Pearson [1] showed that the experimentwise error rate increases when the number of means which are compared is increased. It is thus inappropriate to compare *a posteriori* the basal observations with each of the other observations using the t-test or the Wilcoxon test.

The Dunnett test [2] is a multiple comparison procedure for comparing several treatments with a control. It is frequently used in the kind of experimental set-up described here, but is not appropriate since the observations at different time intervals are made on the same experimental subjects. Analysis of variance for related groups followed by multiple comparisons procedures are better. The parametric analysis of variance for analyzing repeated observations on the same elements is i.e. described by Wiener [3]. Frequency histograms of biomedical data are often non-normal, and outliers are frequently seen. These phenomena may violate the assumptions underlying parametric analysis of variance or reduce its sensitivity. The Friedman test and the Quade test may in these instances perform better than the parametric analogue. These two tests are two representative members of a family of nonparametric tests for randomized complete blocks which i.e. includes the tests of Friedman, Cochran, McNemar and Quade. The Friedman test, which is the multisample extension of the sign test, was invented by the noted economist Milton Friedman in 1937 [4]. The Quade test, which is the multisample extension of the Wilcoxon signed rank test, was invented by Dana Quade 1979 [5], and was named after its inventor by W. J. Conover [6]. Multiple comparisons are performed according to Conover [6]. The present computer program is based on pages 294 to 308 in Conover's text [6] and uses the same labels for the variables as he uses.

STATISTICAL THEORY

The analysis concerns b experimental subjects (blocks) similar to each other in some important aspects subjected to k different treatments or observed at k different time intervals during a treatment. In the Friedman test the ranks within subjects (blocks) $R(X_{ij})$ are found and the sum of the ranks for each treatment R_j where

$$R_j = \sum_{i=1}^{b} R(X_{ij}).$$

The terms $A2$ and $B2$ are then calculated in the following manner;

$$A2 = \sum_{i=1}^{b} \sum_{j=1}^{b} [R(X_{ij})]^2$$

$$B2 = \frac{1}{b} \sum_{j=1}^{k} R_j^2.$$

The test statistic $T2$ is;

$$T2 = \frac{(b-1)\,[B2 - bk(k+1)^2/4]}{A2 - B2}.$$

The null hypothesis is rejected at the level α if $T2$ exceeds the $1-\alpha$ quantile of the F distribution with $k-1$ and $(b-1)(k-1)$ degrees of freedom. If the Friedman test results in rejection of the null hypothesis, the individual observations (treatments) may be compared using the following inequality

$$|R_j - R_i| > t_{1-\alpha/2}\left[\frac{2b\,(A2 - B2)}{(b-1)\,(k-1)}\right]^{1/2}.$$

Where R_j and R_i are the sum of ranks for observations (treatments) i and j respectively and $t_{1-\alpha/2}$ is the $1-\alpha/2$ quantile of the Student's distribution.

The Quade test differs from the Friedman test in that ranks $Q_1, Q_2, \ldots, Q_b$ are assigned to the subjects (blocks) themselves according to the size of the sample range in each block. The block rank Q_i is then multiplied by the difference between the rank within block i, $R(X_{ij})$ and the average rank within blocks, $(k+1)/2$, to get S_{ij} where

$$S_{ij} = Q_i\left[R(X_{ij}) - \frac{k+1}{2}\right]$$

which represents the relative size of each observation within the subject (block), adjusted to reflect the relative significance of the block in which it appears (subject rank weighting). S_j is the sum for each treatment for $j = 1, 2, \ldots, k$.

$$S_j = \sum_{i=1}^{b} S_{ij}.$$

We then calculate the total sum of squares $A1$ and the treatment sum of squares $B1$ in the following manner

$$A1 = \sum_{i=1}^{b} \sum_{j=1}^{k} S_{ij}^2$$

$$B1 = \frac{1}{b} \sum_{j=1}^{k} S_{j\cdot}^2$$

The test statistic $T1$ is then calculated;

$$T1 = \frac{(b-1)\,B1}{A1 - B1}$$

and its significance tested using the F distribution as in the Friedmans test. If this procedure results in rejection of the null hypothesis the following inequality is used for multiple comparisons

$$|S_i - S_j| > t_{1-\alpha/2} \left[\frac{2b\,(A1 - B1)}{(b-1)(k-1)} \right]^{1/2}$$

where $t_{1-\alpha/2}$ is from the t distribution with $(b-1)(k-1)$ degrees of freedom.

PROGRAM DESCRIPTION

The program (see Appendix 2) is written in an elementary subset of BASIC, i.e. which is readily understood and common to most versions of the language. Thus no use is made of structures specific for few computers or statements such as IF . . . THEN . . . ELSE, PRINT USING or ON . . . GOTO. LET is included in all arithmetic statements to indicate that the equal sign in this case means "equal by definition" and since this will always work. However, it may be omitted in most implementations of BASIC. The ATN function is used on line 2740. Some computers use ATAN or ARCTAN for this function. If it is unavailable on your computer, the value may i.e. be approximated using a subroutine published by Lien [7]. Suffix starting at 1 is used when working with arrays, and multiple statements have not been written on the same line. If the present array dimensions are used, the program will compare up to 50 individuals observed at up to 30 occasions i.e. the data matrix as currently dimensioned has room for 50 rows and 30 columns. However, this may be changed by altering the dimension of arrays in line 50. The input and output facilities are very simple in interest of brevity and transportability between different computers.

Parameters and matrices used

$D(50, 30)$	Real array	Matrix of raw data.
$D1(50, 30)$	Real array	Matrix of ranked raw data.
$D2(50, 30)$	Real array	Matrix of S_{ij}.
$B(50)$	Real array	Vector containing S_j.
$R(50)$	Real array	Vector containing R_j.
$N(30, 2)$	Real array	Matrix of: $N(30, 1)$ Sample range in a subject. $N(30, 2)$ Rank of sample range.
$M\%$	Integer	Number of subjects (blocks).
M	Real	Number of subjects (blocks).
$N\%$	Integer	Number of observations (treatments).
T	Real	A temporary variable.
$T1$	Real	Test statistic for overall significance of the Quade test.
$T2$	Real	Test statistic for overall significance of the Friedman test.
$T3$–$T6$	Real	Statistics for multiple comparisons.
S	Real	Smallest observation in a subject.
L	Real	Largest observation in a subject.
$A1$	Real	Test statistic $A1$ for the Quade test.

$A2$	Real	Test statistic $A2$ for the Friedman test.
$B1$	Real	Test statistic $B1$ for the Quade test.
$B2$	Real	Test statistic $B2$ for the Friedman test.
$C1-C9$	Real	Variables in the subroutine to calculate F probability.
$Q1-Q9$	Real	Variables used in the subroutine to calculate t quantiles.
$Y-Y4$	Real	Variables used in the subroutine to calculate t quantiles.
$I\%$, $J\%$, $K\%$	Integer	Loop variables.

Lines 10–50

Following the program presentation, the matrices are dimensioned. If you want to analyze more than 50 subjects (blocks), you should increase the number 50 in all definitions to the number of your choice, and if the number of observations (treatments) made on each subject (blocks) is more or less than 30, the number 30 may be adjusted accordingly.

Lines 70–180

Read the raw data into matrix D(50, 30) from DATA statements located in lines 2820–2950. The number of rows in the present data set is read first ($M\%$) (subjects, blocks), followed by the number of columns ($N\%$) (observations, treatments). Thus every row in the data matrix contains observations from a single subject (block), and each column contains observations (treatments) made at a certain time in relation to the stimulus. The data are read into the matrix one row (subject, block) at a time.

Lines 190–430

Observations (treatments) within subjects (blocks) are ranked and the results are stored in matrix $D1(50, 30)$. Average ranks are used in case of ties. Furthermore, the difference between the largest and smallest observation (treatment) in a subject (block) is calculated, and the result stored in matrix $N(30, 1)$. The algorithm used to perform the ranking process is simple and easy to understand, but may be time consuming in case of large data matrices.

Lines 440–600

Observation ranges in matrix $N(30, 1)$ are ranked, stored in matrix $N(30, 2)$ and printed.

Lines 610–720

Sum of ranks for each treatment R_j are calculated and stored in vector $R(50)$.

Lines 730–830

S_{ij} is calculated and stored in matrix $D3(50, 30)$ as the subject (block) rank $N(I\%, 2)$ multiplied by the difference between the rank within subjects $D1(I\%, J\%)$ and the average rank within observations (treatments) $(N\% + 1)/2$).

Lines 840–1040

Statistic $A1$ for the Quade test is calculated as the sum of squares of S_{ij} calculated over all subjects (blocks) and observations (treatments).

Vector $B(30)$ is calculated as the sum of S_{ij} for each observation (treatment). Statistic $B1$ for the Quade test is calculated as the average sum of squared $B(I\%)$ values summed over all observations (treatments). Thus $B1$ is the "treatment sum of squares".

Lines 1050–1090

The test statistic $T1$ for the Quade test is calculated as $((M\% - 1) * B1)/(A1 - B1)$. The hypothesis that there is no significant change in the observed parameter upon stimulus is rejected at a chosen level if $T1$ exceeds the appropriate quantile of the F distribution with $N\% - 1$ and $(N\% - 1) * (M\% - 1)$ degrees of freedom respectively. The p value associated with the calculated $T1$ value is calculated (using a subroutine starting at line 2270) by means of an algorithm published by J. D. Lee, D. E. Hayes and T. D. Lee [8, 9]. If the overall significance does not reach the $p < 0.05$ level, the program will stop without performing multiple comparisons.

Lines 1100 – 1200

Calculate the appropriate t quantiles associated with $p < 0.05, p < 0.01$ and $p < 0.001$ and $(N\% - 1)\ (M\% - 1)$ degrees of freedom ($Q2$) ($T3$, $T4$ and $T5$ respectively). The subroutine starting at line 1960 is based on ACM algorithm 396 [10].

Lines 1210–1440

Multiple comparisons are performed by test parameter $T6$ which is equal to $SQR((2 * M\% * (A1 - B1))/((M\% - 1) * (N\% - 1)))$ multiplied by the appropriate t values. The difference between sums is calculated as the absolute value of the difference between the $B(I\%)$ value for one observation (treatment) and the $B(J\%)$ value for another observation (treatment) and compared to the $T6$ value.

Lines 1450–1640

Calculate the $A2$ and $B2$ values for the Friedman test.

Lines 1650–1690

Calculate the $T2$ test statistic for the Friedman test and its associated significance using the F subroutine starting at line 2290.

Lines 1700–1940

Perform multiple comparisons on the data if Friedman test rejects the null hypothesis.

RESULTS

Test data (see Appendix 1) stem from experiments where release of tachykinins from tumors into the blood stream of 12 carcinoid patients was stimulated by means of pentagastrin [11]. Since the basal plasma concentrations of tachykinins were very different among the patients, the change in plasma concentrations was expressed as percent increase (+) or decrease (−) in relation to a basal value from the same patient obtained 15 min before the stimulus. Another basal sample obtained shortly before the stimulus was used to get an impression of the spontaneous variation in the plasma concentrations. The first column in the list of raw data shows that the spontaneous variation in the basal plasma levels of tachykinins is small. Seven other plasma samples were obtained at the same points in time after the stimulus. The plasma concentrations of tachykinins increased in all patients upon pentagastrin stimulation.

The present version the program prints out the original data and all relevant intermediate results. A numerical routine is used to calculate the significance (p value) of the test parameters $T1$ and $T2$ for the overall test of the null hypothesis that there are "no treatment differences". In the case of the test data (Appendix 1) $T1$ equal to 11.89 gives $p < 0.001$ for the Quade test. The program will therefore proceed with the multiple comparisons procedures. Generally when the p value is larger than 0.05, the null hypothesis is accepted and the program modules terminate without performing multiple comparisons. However, if the null hypothesis is rejected ($p < 0.05$) the computer tests all pairs of observations (treatments) with each other. The probability levels [ns (not significant), $p < 0.05$, $p < 0.01$ and $p < 0.001$] given in the multiple comparisons procedure have no real probabilistic meaning since they depend on the results of the first stage overall test. However, they have many characteristics

Table 1. Relation between the number of subjects in the sample data and the overall probabilities in the Quade and Friedman tests as calculated by the present program

Number of subjects	Quade test	Friedman test
5	p = 3.54E-03	p = 7.87E-05
4	p = 0.018	p = 0.002
3	p = 0.011	p = 0.001

of the procedures usually used to compare two samples and most often contain the information that the experimenter really wants, since he is usually not as interested in knowing whether any differences among the treatments exists as he is in knowing which observations (treatments) are different.

Conover [6] indicates that the Quade test is more powerful than the Friedman test for small number (less than five) of treatments while the Friedman test is more powerful than the Quade test when the number of treatments is five or more. This general rule is probably not always true. Thus, if the number of subjects in the present data is systematically reduced from the top, down, the overall probability for the two tests changes as shown in Table 1. Furthermore, the reader will observe the same phenomenon by analyzing the data on respectively page 298 and page 301 in Conover's book [6].

SUMMARY

A computer program written in an elementary subset of BASIC is described which performs the Friedman and the Quade tests followed by multiple comparisons. The Friedman and Quade tests with multiple comparison procedures as described by Conover are nonparametric two-way analyses of variance which may be used to compare several related samples. The program may i.e. be used to analyze biomedical data where the response of experimental subjects to a stimulus is monitored at time intervals. It gives the relevant intermediate results and includes subroutines to calculate probability (p) values both for the overall significance and for the multiple comparisons.

Acknowledgements—The present study was supported by the Swedish Medical Research Council (03X-07464), Konung Gustav V:s jubileumsfond (85:569), The Swedish Cancer Research Fund (2313-B87-01X), Funds of the Karolinska Institute, Åke Vibergs Stiftelse, Nordisk Insulinfond, Smith Kline and French and The Swedish Society for Medical Sciences.

REFERENCES

1. E. S. Pearson, The probability integral of the range in samples of n observations from a normal population, *Biometrica* **32,** 301–308 (1942).
2. C. W. Dunnett, A multiple comparison procedure for comparing several treatments with a control, *J. Am. Stat. Assoc.* **50,** 1096–1121 (1955).
3. B. J. Winer, Single-factor experiments having repeated measures on the same elements, *Statistical Principles In Experimental Design,* 2nd Edn, Chapter 4. McGraw-Hill, New York (1971).
4. M. Friedman, The use of ranks to avoid the assumption of normality implicit in the analysis of variance, *J. Am. Stat. Assoc.* **32,** 675–701 (1937).
5. D. Quade, Using weighted rankings in the analysis of complete blocks with additive block effects, *J. Am. Stat. Assoc.* **74,** 680–683 (1979).
6. C. W. Conover, *Practical Nonparametric Statistics,* 2nd Edn. John Wiley, New York (1980).
7. D. A. Lien, *The BASIC Handbook.* Compusoft, Inc. (1981).
8. J. D. Lee and D. G. Hayes, Some elementary statistical and numerical procedures—an interactive approach. *Comput. Educ.* **3,** 145–176 (1978).
9. J. D. Lee and T. D. Lee, *Statistics and Numerical Methods in BASIC for Biologists.* Van Nostrand Reinhold, New York (1982).
10. G. W. Hill, Algorithm 396, Student's t-quantiles. Collected Algorithms from CACM. *Comm. ACM* **13,** 617–619 (1970).
11. I. Norheim, E. Theodorsson-Norheim, E. Brodin and K. Öberg, Tachykinins in carcinoid tumors: their use as a tumor marker and possible role in the carcinoid flush. *J. clin. Endocrinol. Metab.* **63,** 605–612 (1986).

APPENDIX 1

PROGRAM OUTPUT WHEN ANALYSING THE TEST DATA

```
***********************************************************
*  FRIEDMAN AND QUADE TESTS FOR SEVERAL RELATED SAMPLES  *
***********************************************************

A list of your raw data

-10  13  42  28  41  31  9  0
 2  6  210  398  235  198  99  73
 6  35  403  270  251  117  44  21
-13 -29  344  260  161  177 -79 -81
 0  0  729  579  596  386  318  300
 0  51  47  27  1  19 -3 -1
 0  6  18 -3 -24  10 -9  1
 1  17  50  65  34 -1 -28 -1
 17  159  72  50  28  34  37 -1
 9  80  148  146  84 -1  64  35
 29  29  226  298  148  137  92  71
 17 -19  71  166  78 -24  8 -8

A list of your ranked data

 1  4  8  5  7  6  3  2  Sample range =  52
 1  2  6  8  7  5  4  3  Sample range =  396
 1  3  8  7  6  5  4  2  Sample range =  397
 4  3  8  7  5  6  2  1  Sample range =  425
 1.5  1.5  8  6  7  5  4  3  Sample range =  729
 3  8  7  6  4  5  1  2  Sample range =  54
 4  6  8  3  1  7  2  5  Sample range =  42
 4  5  7  8  6  2.5  1  2.5  Sample range =  93
 2  8  7  6  3  4  5  1  Sample range =  160
 2  5  8  7  6  1  4  3  Sample range =  149
 1.5  1.5  7  8  6  5  4  3  Sample range =  269
 5  2  6  8  7  1  4  3  Sample range =  190

Rank of sample ranges

 2  9  10  11  12  3  1  4  6  5  8  7

Sum of ranks for each treatment

 30  49  88  79  65  52.5  38  30.5

List of Sij

-7 -1  7  1  5  3 -3 -5
-31.5 -22.5  13.5  31.5  22.5  4.5 -4.5 -13.5
-35 -15  35  25  15  5 -5 -25
-5.5 -16.5  38.5  27.5  5.5  16.5 -27.5 -38.5
-36 -36  42  18  30  6 -6 -18
-4.5  10.5  7.5  4.5 -1.5  1.5 -10.5 -7.5
-.5  1.5  3.5 -1.5 -3.5  2.5 -2.5  .5
-2  2  10  14  6 -8 -14 -8
-15  21  15  9 -9 -3  3 -21
-12.5  2.5  17.5  12.5  7.5 -17.5 -2.5 -7.5
-24 -24  20  28  12  4 -4 -12
 3.5 -17.5  10.5  24.5  17.5 -24.5 -3.5 -10.5
*** Results from the Quade test ***
```

```
The Quade test Al =  27188
The Quade test Bl =  14122.2

The Quade test Tl =  11.8893      p =  1.07288E-06

Difference between sums for Quade test

Observation  1
         -observation 1  =  0      NS
         -observation 2  =  75     NS
         -observation 3  =  390    p<0.001
         -observation 4  =  364    p<0.001
         -observation 5  =  277    p<0.001
         -observation 6  =  160    p<0.01
         -observation 7  =  90     NS
         -observation 8  =  4      NS
Observation  2
         -observation 1  =  75     NS
         -observation 2  =  0      NS
         -observation 3  =  315    p<0.001
         -observation 4  =  289    p<0.001
         -observation 5  =  202    p<0.01
         -observation 6  =  85     NS
         -observation 7  =  15     NS
         -observation 8  =  71     NS
Observation  3
         -observation 1  =  390    p<0.001
         -observation 2  =  315    p<0.001
         -observation 3  =  0      NS
         -observation 4  =  26     NS
         -observation 5  =  113    p<0.05
         -observation 6  =  230    p<0.001
         -observation 7  =  300    p<0.001
         -observation 8  =  386    p<0.001
Observation  4
         -observation 1  =  364    p<0.001
         -observation 2  =  289    p<0.001
         -observation 3  =  26     NS
         -observation 4  =  0      NS
         -observation 5  =  87     NS
         -observation 6  =  204    p<0.01
         -observation 7  =  274    p<0.001
         -observation 8  =  360    p<0.001
Observation  5
         -observation 1  =  277    p<0.001
         -observation 2  =  202    p<0.01
         -observation 3  =  113    p<0.05
         -observation 4  =  87     NS
         -observation 5  =  0      NS
         -observation 6  =  117    p<0.05
         -observation 7  =  187    p<0.01
         -observation 8  =  273    p<0.001
Observation  6
         -observation 1  =  160    p<0.01
         -observation 2  =  85     NS
         -observation 3  =  230    p<0.001
         -observation 4  =  204    p<0.01
         -observation 5  =  117    p<0.05
         -observation 6  =  0      NS
         -observation 7  =  70     NS
         -observation 8  =  156    p<0.01
```

```
Observation  7
          -observation 1  =  90     NS
          -observation 2  =  15     NS
          -observation 3  =  300    p<0.001
          -observation 4  =  274    p<0.001
          -observation 5  =  187    p<0.01
          -observation 6  =  70     NS
          -observation 7  =  0      NS
          -observation 8  =  86     NS
Observation  8
          -observation 1  =  4      NS
          -observation 2  =  71     NS
          -observation 3  =  386    p<0.001
          -observation 4  =  360    p<0.001
          -observation 5  =  273    p<0.001
          -observation 6  =  156    p<0.01
          -observation 7  =  86     NS
          -observation 8  =  0      NS

*** Results from the Friedman test ***

The Friedman test A2 =  2446.5
The Friedman test B2 =  2220.13

The Friedman test T2 =  13.4175      p =  8.34465E-07

Difference between sums for Friedman test

Observation  1
          -observation 1  =  0      NS
          -observation 2  =  19     p<0.05
          -observation 3  =  58     p<0.001
          -observation 4  =  49     p<0.001
          -observation 5  =  35     p<0.001
          -observation 6  =  22.5   p<0.01
          -observation 7  =  8      NS
          -observation 8  =  .5     NS
Observation  2
          -observation 1  =  19     p<0.05
          -observation 2  =  0      NS
          -observation 3  =  39     p<0.001
          -observation 4  =  30     p<0.001
          -observation 5  =  16     p<0.05
          -observation 6  =  3.5    NS
          -observation 7  =  11     NS
          -observation 8  =  18.5   p<0.05
Observation  3
          -observation 1  =  58     p<0.001
          -observation 2  =  39     p<0.001
          -observation 3  =  0      NS
          -observation 4  =  9      NS
          -observation 5  =  23     p<0.01
          -observation 6  =  35.5   p<0.001
          -observation 7  =  50     p<0.001
          -observation 8  =  57.5   p<0.001
Observation  4
          -observation 1  =  49     p<0.001
          -observation 2  =  30     p<0.001
          -observation 3  =  9      NS
          -observation 4  =  0      NS
          -observation 5  =  14     p<0.05
```

```
         -observation 6  =  26.5  p<0.01
         -observation 7  =  41    p<0.001
         -observation 8  =  48.5  p<0.001
Observation 5
         -observation 1  =  35    p<0.001
         -observation 2  =  16    p<0.05
         -observation 3  =  23    p<0.01
         -observation 4  =  14    p<0.05
         -observation 5  =  0     NS
         -observation 6  =  12.5  NS
         -observation 7  =  27    p<0.001
         -observation 8  =  34.5  p<0.001
Observation 6
         -observation 1  =  22.5  p<0.01
         -observation 2  =  3.5   NS
         -observation 3  =  35.5  p<0.001
         -observation 4  =  26.5  p<0.01
         -observation 5  =  12.5  NS
         -observation 6  =  0     NS
         -observation 7  =  14.5  p<0.05
         -observation 8  =  22    p<0.01
Observation 7
         -observation 1  =  8     NS
         -observation 2  =  11    NS
         -observation 3  =  50    p<0.001
         -observation 4  =  41    p<0.001
         -observation 5  =  27    p<0.001
         -observation 6  =  14.5  p<0.05
         -observation 7  =  0     NS
         -observation 8  =  7.5   NS
Observation 8
         -observation 1  =  .5    NS
         -observation 2  =  18.5  p<0.05
         -observation 3  =  57.5  p<0.001
         -observation 4  =  48.5  p<0.001
         -observation 5  =  34.5  p<0.001
         -observation 6  =  22    p<0.01
         -observation 7  =  7.5   NS
         -observation 8  =  0     NS
```

APPENDIX 2

```
10 PRINT "************************************************************"
20 PRINT "*  FRIEDMAN AND QUADE TESTS FOR SEVERAL RELATED SAMPLES  *"
30 PRINT "************************************************************"
40 PRINT
50 DIM N(30,2),B(50),D(50,30),D1(50,30),D2(50,30),R(50)
60 REM ------------------ READ IN THE RAW DATA --------------------
70 PRINT "A list of your raw data"
80 PRINT
90 READ M%
100 READ N%
110 FOR I%=1 TO M%
120     FOR J%=1 TO N%
130             READ D(I%,J%)
140             PRINT D(I%,J%);
150     NEXT J%
160     PRINT
170 NEXT I%
180 PRINT
```

```
190 REM ---- RANK DATA WITHIN SUBJECTS AND STORE IN D1(50,30) -----
200 REM ------- CALCULATE SAMPLE RANGE AND STORE IN N(I,2) --------
210 PRINT "A list of your ranked data"
220 PRINT
230 FOR I%=1 TO M%
240     LET S=D(I%,1)
250     LET L=D(I%,1)
260     FOR J%=1 TO N%
270             IF S>D(I%,J%) THEN S=D(I%,J%)
280             IF L<D(I%,J%) THEN L=D(I%,J%)
290             LET T=.5
300             FOR K%=1 TO N%
310                     IF D(I%,K%)>D(I%,J%) THEN GOTO 360
320                     IF D(I%,K%)=D(I%,J%) THEN GOTO 350
330                     LET T=T+1
340                     GOTO 360
350                     LET T=T+.5
360             NEXT K%
370             LET D1(I%,J%)=T
380             PRINT D1(I%,J%);
390     NEXT J%
400     LET N(I%,1)=L-S
410     PRINT " Sample range = ";N(I%,1)
420 NEXT I%
430 PRINT
440 REM ------ RANK OBSERVATION RANGES AND STORE IN N(I%,2) -------
450 PRINT "Rank of sample ranges "
460 PRINT
470 FOR I%=1 TO M%
480   LET T=.5
490     FOR J%=1 TO M%
500             IF N(J%,1)>N(I%,1) THEN GOTO 550
510             IF N(J%,1)=N(I%,1) THEN GOTO 540
520             LET T=T+1
530             GOTO 550
540             LET T=T+.5
550     NEXT J%
560     LET N(I%,2)=T
570     PRINT N(I%,2);
580 NEXT I%
590 PRINT
600 PRINT
610 REM --------- FIND THE SUM OF RANKS FOR EACH TREATMENT --------
620 PRINT "Sum of ranks for each treatment "
630 PRINT
640 FOR J%=1 TO N%
650     LET R(J%)=0
660     FOR I%=1 TO M%
670             LET R(J%)=R(J%)+D1(I%,J%)
680     NEXT I%
690     PRINT R(J%);
700 NEXT J%
710 PRINT
720 PRINT
730 REM ------------------- CALCULATE Sij ----------------------
740 PRINT "List of Sij"
750 PRINT
760 FOR I%=1 TO M%
770     FOR J%=1 TO N%
780         LET D2(I%,J%)=N(I%,2)*(D1(I%,J%)-((N%+1)/2))
```

```
790         PRINT D2(I%,J%);
800     NEXT J%
810     PRINT
820 NEXT I%
830 PRINT
840 REM ------------ CALCULATE A1 FOR THE QUADE TEST -------------
850 PRINT "*** Results from the Quade test ***
860 PRINT
870 LET A1=0
880 FOR I%=1 TO M%
890     FOR J%=1 TO N%
900         LET A1=A1+D2(I%,J%)*D2(I%,J%)
910     NEXT J%
920 NEXT I%
930 PRINT "The Quade test A1 = ";A1
940 REM ------------ CALCULATE B1 FOR THE QUADE TEST -------------
950 LET B1=0
960 FOR I%=1 TO N%
970     FOR J%=1 TO M%
980         LET B(I%)=B(I%)+D2(J%,I%)
990     NEXT J%
1000    LET B1=B1+B(I%)*B(I%)
1010 NEXT I%
1020 LET B1=B1/M%
1030 PRINT "The Quade test B1 = ";B1
1040 PRINT
1050 REM ------------ CALCULATE T1 FOR THE QUADE TEST -------------
1060 LET T1=((M%-1)*B1)/(A1-B1)
1070 PRINT "The Quade test T1 = ";T1;
1080 LET T9=T1
1090 GOSUB 2270
1100 REM ----------- CALCULATE T DISTRIBUTION QUANTILES ----------
1110 LET Q1=.05
1120 LET Q2=(N%-1)*(M%-1)
1130 GOSUB 1960
1140 LET T3=T
1150 LET Q1=.01
1160 GOSUB 1960
1170 LET T4=T
1180 LET Q1=.001
1190 GOSUB 1960
1200 LET T5=T
1210 REM ----- DO NOT PERFORM MULTIPLE COMPARISONS IF OVERALL ----
1220 REM ----- TEST IS NOT SIGNIFICANT                         ----
1230 IF P>.05 THEN GOTO 1460
1240 LET T6=SQR((2*M%*(A1-B1))/((M%-1)*(N%-1)))
1250 PRINT
1260 PRINT "Difference between sums for Quade test"
1270 PRINT
1280 FOR I%=1 TO N%
1290     PRINT "Observation ";I%
1300      FOR J%=1 TO N%
1310         LET E=ABS(B(I%)-B(J%))
1320         PRINT TAB(10);"-observation";J%;" = ";E;
1330         IF E>T6*T5 THEN GOTO 1380
1340         IF E>T6*T4 THEN GOTO 1400
1350         IF E>T6*T3 THEN GOTO 1420
1360         PRINT TAB(35);"NS"
1370         GOTO 1430
1380         PRINT TAB(35);"p<0.001"
```

```
1390          GOTO 1430
1400          PRINT TAB(35);"p<0.01"
1410          GOTO 1430
1420          PRINT TAB(35);"p<0.05"
1430       NEXT J%
1440 NEXT I%
1450 REM ---------- CALCULATE A2 FOR THE FRIEDMAN TEST -----------
1460 PRINT
1470 PRINT "*** Results from the Friedman test ***"
1480 PRINT
1490 LET A2=0
1500 FOR I%=1 TO M%
1510    FOR J%=1 TO N%
1520        LET A2=A2+D1(I%,J%)*D1(I%,J%)
1530    NEXT J%
1540 NEXT I%
1550 PRINT "The Friedman test A2 = ";A2
1560 REM ---------- CALCULATE B2 FOR THE FRIEDMAN TEST -----------
1570 LET B2=0
1580 FOR I%=1 TO N%
1590    LET B2=B2+R(I%)*R(I%)
1600 NEXT I%
1610 LET M=M%
1620 LET B2=B2/M
1630 PRINT "The Friedman test B2 = ";B2
1640 PRINT
1650 REM ---------- CALCULATE T2 FOR THE FRIEDMAN TEST -----------
1660 LET T2=((M%-1)*(B2-(M%*N%*((N%+1)*(N%+1))/4)))/(A2-B2)
1670 PRINT "The Friedman test T2 = ";T2;
1680 LET T9=T2
1690 GOSUB 2270
1700 REM ---- DO NOT PERFORM MULTIPLE COMPARISONS IF OVERALL -----
1710 REM ---- TEST IS NOT SIGNIFICANT                        -----
1720 IF P>.05 THEN GOTO 1940
1730 LET T6=SQR((2*M%*(A2-B2))/((M%-1)*(N%-1)))
1740 PRINT
1750 PRINT "Difference between sums for Friedman test"
1760 PRINT
1770 FOR I%=1 TO N%
1780     PRINT "Observation ";I%
1790     FOR J%=1 TO N%
1800         LET E=ABS(R(I%)-R(J%))
1810         PRINT TAB(10);"-observation";J%;" = ";E;
1820         IF E>T6*T5 THEN GOTO 1870
1830         IF E>T6*T4 THEN GOTO 1890
1840         IF E>T6*T3 THEN GOTO 1910
1850         PRINT TAB(35);"NS"
1860         GOTO 1920
1870         PRINT TAB(35);"p<0.001"
1880         GOTO 1920
1890         PRINT TAB(35);"p<0.01"
1900         GOTO 1920
1910         PRINT TAB(35);"p<0.05"
1920     NEXT J%
1930 NEXT I%
1940 END
1950 REM --------- SUBROUTINE TO CALCULATE T QUANTLES -----------
1960 IF Q2>2 THEN GOTO 2000
1970 IF Q2=1 THEN T=COS(Q1*1.5708)/SIN(Q1*1.5708)
1980 IF Q2=2 THEN T=SQR(2/(Q1*(2-Q1))-2)
```

```
1990 RETURN
2000 LET Q4=1/(Q2-.5)
2010 LET Q5=48/Q4^2
2020 LET Q6=((20700*Q4/Q5-98)*Q4-16)+96.36
2030 LET Q7=((94.5/(Q5+Q6)-3)/Q5+1)*SQR(Q4*1.5708)*Q2
2040 LET Q8=Q7*Q1
2050 LET Y=Q8^(2/Q2)
2060 IF Y>.05+Q4 THEN GOTO 2080
2070 GOTO 2190
2080 LET Q9=SQR(-2*LOG(Q1))
2090 LET Q8=2.51552+Q9*(.802853+.010328*Q9)
2100 LET Q8=Q9-Q8/(1+Q9*(1.43279+Q9*(.189269+.001308*Q9)))
2110 LET Y=(2*Q8)^2
2120 IF Q2<5 THEN LET Q6=Q6+.3*(Q2-4.5)*(Q8+.6)
2130 LET Q6=(((.05*Q7*Q8-5)*Q8-7)*Q8-2)*Q8+Q5+Q6
2140 LET Y=(((((.4*Y+6.3)*Y+36)*Y+94.5)/Q6-Y-3)/Q5+1)*Q8
2150 LET Y=Q4*Y^2
2160 IF Y>.002 THEN LET Y=EXP(Y)-1
2170 IF Y<=.002 THEN LET Y=.5*Y^2+Y
2180 GOTO 2240
2190 LET Y1=(Q2+2)*3
2200 LET Y2=Q2+4
2210 LET Y3=Q2+1
2220 LET Y4=Q2+6
2230 LET Y=((1/((Y4/(Q2*Y)-.089*Q7-.822)*Y1)+.5/Y2)*Y-1)*Y3/(Q2+2)+1/
2240 LET T=SQR(Q2*Y)
2250 RETURN
2260 REM --------- SUBROUTINE TO CALCULATE F PROBABILITY ----------
2270 LET C3=0
2280 LET C1=N%-1
2290 LET C2=(N%-1)*(M%-1)
2300 IF C1=2*INT(C1/2+.1) THEN GOTO 2330
2310 IF C2=2*INT(C2/2+.1) THEN GOTO 2370
2320 GOTO 2530
2330 LET C4=1/(1+C2/(T9*C1))
2340 LET C5=C1+1
2350 LET C6=C2-2
2360 GOTO 2410
2370 LET C3=1
2380 LET C4=1/(1+T9*C1/C2)
2390 LET C5=C2+1
2400 LET C6=C1-2
2410 LET C7=0
2420 LET C8=1
2430 LET C9=2
2440 LET C7=C7+C8
2450 LET C8=C8*C4*(C9+C6)/C9
2460 LET C9=C9+2
2470 IF C9<C5 THEN GOTO 2440
2480 IF C3=0 THEN GOTO 2510
2490 LET P=100*C7*((SQR(1-C4))^C1)
2500 GOTO 2740
2510 LET P=100*C7*((SQR(1-C4))^C2)
2520 GOTO 2740
2530 LET C4=1/(1+T9*C1/C2)
2540 LET C7=0
2550 LET C8=1
2560 LET C9=2
2570 LET C5=C2
2580 GOTO 2620
```

```
2590 LET C7=C7+C8
2600 LET C8=C8*C4*C9/(C9+1)
2610 LET C9=C9+2
2620 IF C9<C5 THEN GOTO 2590
2630 LET C8=C8*C2
2640 LET C9=3
2650 LET C5=C1+1
2660 LET C6=C2-2
2670 GOTO 2710
2680 LET C7=C7-C8
2690 LET C8=C8*(1-C4)*(C9+C6)/C9
2700 LET C9=C9+2
2710 IF C9<C5 THEN GOTO 2680
2720 LET T9=ATN(SQR(T9*C1/C2))
2730 LET P=100*(1-2*(T9+C7*SQR((1-C4)*C4))/3.14159)
2740 IF P <= 50 THEN GOTO 2760
2750 LET P=100-P
2760 LET P=ABS(P/100)
2770 PRINT "    p = ";P
2780 RETURN
2790 REM
2800 REM ---------------------- RAW DATA ------------------------
2810 REM
2820 DATA 12
2830 DATA 8
2840 DATA -10,13,42,28,41,31,9,0
2850 DATA 2,6,210,398,235,198,99,73
2860 DATA 6,35,403,270,251,117,44,21
2870 DATA -13,-29,344,260,161,177,-79,-81
2880 DATA 0,0,729,579,596,386,318,300
2890 DATA 0,51,47,27,1,19,-3,-1
2900 DATA 0,6,18,-3,-24,10,-9,1
2910 DATA 1,17,50,65,34,-1,-28,-1
2920 DATA 17,159,72,50,28,34,37,-1
2930 DATA 9,80,148,146,84,-1,64,35
2940 DATA 29,29,226,298,148,137,92,71
2950 DATA 17,-19,71,166,78,-24,8,-8
```

About the Author — ELVAR THEODORSSON-NORHEIM was born in Iceland, 1953. He attended Medical School in Iceland, and got an MD licence in 1980. From 1980 to 1983 he did neuropeptide research at the Department of Pharmacology, Karolinska Institute, Stockholm. His thesis dealt with biochemical and clinical aspects of neurotensin. Since 1983 he has been working with clinical, biochemical and physiological aspects of peptide transmitters at the Department of Clinical Chemistry, Karolinska Hospital, Stockholm, where he became Assistant Professor in 1986.